HOW08 - 2008

The Plot Thickens from PLOT to GPLOT

Wendi L. Wright, CTB/McGraw-Hill, Harrisburg, PA

ABSTRACT

This paper starts with a look at basic plotting using PROC PLOT. A dataset with the daily number of hits on three websites is used. The presentation continues with examples of how to create more informative and professional looking plots by adding titles, footnotes, and reference lines and changing the symbols, legends and axes. This is done in a step by step approach with examples of how the plot changes at each step.

A look at PROC GPLOT is next using the same Web example. The final plot created using PROC PLOT is compared with the same program run in the PROC GPLOT procedure. Further improvements are made in the same step by step fashion using various SAS Graph® statements. The first change is to the titles and tootnotes. Then, one at a time, changes are made to the symbols (SYMBOL statement), the legend (LEGEND statement), and the axes (AXIS statement).

INTRODUCTION

The data to be used in the plots are in a permanent SAS dataset that contains daily counts of hits on each of 3 websites. The SAS® dataset called 'hits' is sorted by date and has the following variables.


- Date format yymmdd8.
- Day day of month in \$char2. format
- Web1 hits on Website 1
- Web2 hits on Website 2
- Web3 hits on Website 3
- Total total number of hits on all three websites

PROC PLOT

TIP 1 - REVIEW the DATA!

The data is what eventually drives the options you select. It is a good idea to begin with a review of the data before deciding what options will be needed. Using PROC PLOT, a program was created to analyze the data above. The analysis looks at the total number of hits on all websites.

PROC PLOT DATA=perm.hits	s;
PLOT total * date;	
RUN:	




TIP 2 – Use Descriptive TITLES and FOOTNOTES!


Titles should always explain the plot and the data included. Footnotes can be used to indicate the date the plot was produced and ownership of the data. Add titles using the SAS TITLE statements. Always have titles that fully describe the plot. To remove the SAS date at the top of the plot, use the NODATE option on the OPTIONS statement.

date

```
OPTIONS NODATE;
TITLE 'Total Number of Hits on Websites 1, 2, and 3';
TITLE2 'For the Month of August 2007';
TITLE3 'Figure 2';
FOOTNOTE 'Company Name 9/15/07';
PROC PLOT DATA=perm.hits;
PLOT total * date;
RUN;
```



Plot of total*date. Legend: A = 1 obs, B = 2 obs, etc.



Company Name 9/15/07

TIP 3 - Choose SYMBOLS Carefully!

Symbols should be easy to see, understand, and to distinguish from one another. PROC PLOT uses letters to show how many observations are at each plot point. In the above plot, there is only one observation at each point. The letters that show how many observations are at each point in the plot are, therefore, not needed and the use of a symbol makes the plot design cleaner. To change the symbol to an asterisk, add the text ' = '*' ' after the plot variables in the PLOT statement.


PROC PLOT DATA=perm.hits;

PLOT total * date = '*';

Run;

Total Number of Hits on Websites 1, 2, and 3
For the Month of August 2007
Figure 3

Plot of total*date. Symbol used is '*'.




date

Company Name 9/15/07

TIP 4 - LEGENDS Only When Needed!

Only use legends if they add value to your plot (i.e. explain something that is not otherwise clearly understood). In this example, since the legend at the top is not necessary for understanding the plot, it was removed using the NOLEGEND option in the PROC PLOT statement.

```
PROC PLOT DATA=perm.hits NOLEGEND;
PLOT total * date = '*';
RUN;
```





TIP 5 - Clear AXIS Labels!

Axis labels and values should be easy to see, understand and clearly describe what the axis represents. For example, if the tick marks represent cities on one axis, it's better to label each tick mark with the city name rather than use a label like 'City 1'. Labeling the axis in this way avoids the need for a legend. To label the variables on the axes, use the label statement in the data step. Because the text in the Date axis repeats the month and year, I decided to use the DAY variable instead. The labeling of the variables was done in a data step.

The code to implement Tip 5 is below. Note: a label was also added for the variable web1. You will see the reason for this in Tip 7.

```
DATA perm.hits;
SET perm.hits;
LABEL Day='Day of Month in August';
LABEL Total='Total Number of Hits';
LABEL web1='Number of Hits';
RUN;
PROC PLOT DATA=perm.hits NOLEGEND;
PLOT total * day = '*';
RUN;
```







Company Name 9/15/07

TIP 6 - REFERENCE Lines!

Use reference lines to add meaningful information to your plot. In this example, an e-mail was sent out to advertise the use of one of the websites on August 17th. In order to explain the change in volume, a reference line was added to indicate the e-mail marketing and the resulting spike in volume. Use the HREF option on the PLOT statement to get a horizontal line.

PROC PLOT DATA=perm.hits NOLEGEND;
PLOT total * day = '*' / HREF='17';
RUN;





Day of Month in August

Company Name 9/15/07

TIP 7 - Use More Detail!

If the information is available, add as much detail to your plot as you can. The data for each of the three websites is available separately. Displaying the three sites' counts separately (rather than just the total count) clarifies what is really happening with the data. To do this use the OVERLAY option on the PLOT statement.

The data for each website should have a separate symbol. Using the same technique listed above (in Tip 3), I chose a different symbol for each data source. Now that each website has its own symbol, a legend is needed and the NOLEGEND option was removed.

The first variable plotted on an axis is the variable that is used to label the axis. Because the variable 'Web1' (see Tip 5 for assignment of the label) was listed first on the horizontal axis, the label for this variable is the label that is printed on the plot.

Below is the code to implement these changes. The TITLE statement was changed slightly as well.

```
TITLE 'Number of Hits on Websites1, 2, and 3';

TITLE2 'For the Month of August 2007';

TITLE3 'Figure 7';

PROC PLOT DATA=perm.hits;


PLOT Web1 * day = '*'

Web2 * day = '+'

Web3 * day = '-'

/ OVERLAY HREF = '17';

RUN;
```



TIP 8 - Add a FRAME!

A frame makes it easier to line up where the points are within the plot. For PROC PLOT, it shows tick marks at the upper and right sides of the plot. Using the BOX option on the PLOT statement, a box was drawn around the plot.

```
PROC PLOT DATA=perm.hits;

PLOT Web1 * day = '*'

Web2 * day = '+'

Web3 * day = '-'

/ OVERLAY HREF = '17' BOX;

RUN;
```

```
Number of Hits on Websites 1, 2, and 3
                        For the Month of August 2007
                                  Figure 8
                  Plot of web1*day. Symbol used is '*'.
                  Plot of web2*day. Symbol used is '+'.
                  Plot of web3*day. Symbol used is '-'.
u
  1500
m
b
e
r
  1000
o
f
   500
Н
i
t
     0
              0 0 0 0 0 0 0 1 1 1
                                     1
                              Day of Month in August
```

NOTE: 2 obs hidden.

Company Name 9/15/07

FINAL PROC PLOT PROGRAM

```
DATA perm.hits;
       SET perm.hits;
       Day=substr (date,7,2);
       Label Day='Day of Month in August';
       Label Total='Total Number of Hits';
       Label web1='Number of Hits';
   RUN;
   TITLE 'Number of Hits on Websites 1, 2, and 3';
   TITLE2 'For the Month of August 2007':
   FOOTNOTE 'Company Name 9/15/07';
   PROC PLOT DATA=perm.hits;
       PLOT Web1 * day = '*'
           Web2 * day = '+'
           Web3 * day = '-'
           /OVERLAY HREF = '17' BOX;
RUN;
```

FINAL NOTES ON PROC PLOT

There are other options that can also be used with the plot statement. There are options that control placement of labels on the plot. Other options that may interest you are:

- HREFCHAR and VREFCHAR these two options allow you to specify the character to use when drawing reference lines.
- HREVERSE and VREVERSE these options reverse the order of the values on the axes.
- HZERO and VZERO these options assign a value of zero to the first tick mark on the axes.
- UNIFORM standardizes the x and y axis values across by groups when producing multiple plots.

It is also possible to label plot points on the plot. By adding a \$ and a label-variable name to the plot statement, PROC PLOT will use the values to place labels on your plot. Below is the syntax of the code.

```
PROC PLOT DATA=mydata;
PLOT y * x $ label_variable / <other options>;
RUN;
```

The label placements can be adjusted by using the placement and penalty options. See the SAS documentation for more information on this.


PROC GPLOT

STEP 1 - Comparing Plot and GPlot

A review of the options used in the PLOT statement in the PROC PLOT program determined that the BOX option is not available for use in GPLOT, but the FRAME option is and they do very similar things. Note: another option that is available to GPLOT and not PLOT is the GRID option, which causes reference lines to be drawn at every major tick mark. However, a full grid on this plot would make the plot much too busy, so this option was not used. The OVERLAY and HREF options are valid in both PLOT and GPLOT, so those options were kept. Below is the code used to run the PROC GPLOT program with the change to the options on the PLOT statement from BOX to FRAME.

```
TITLE 'Number of Hits on Websites 1, 2, and 3';
TITLE2 'For the Month of August 2007';
TITLE3 'Figure 9';
PROC GPLOT DATA=perm.hits;
PLOT Web1 * day = '*'
Web2 * day = '+'
Web3 * day = '-'
/ OVERLAY HREF = '17' FRAME;
RUN;
```





STEP 2 - Modifying the Titles and Footnotes

PROC GPLOT allows changes to font, height, and color of the text. PROC GPLOT also allows the placement of the title to be altered. Using PROC GPLOT, a box can be drawn around the title or the title can be underlined. Note these options are only available for certain SAS Graph procedures and are not available with PROC PLOT. All these options work for text in titles, notes, and footnotes.

OPTION	WHAT IT DOES		
COLOR =	Color of the text		
BCOLOR =	Background color if using the box option		
FONT =	Font of the text		
HEIGHT =	Height of the letters		
JUSTIFY = Left Center Right	Justifies portions of the text		
ANGLE = degrees	Turns the entire line of text the number of degrees specified		
ROTATE = degrees	Turns each letter in the text the number of degrees specified		
BOX = 1 2 3 4	Creates a box around the text -1 lightest line, $4 =$ darkest line		
DRAW = (coordinate pairs)	Draws lines between the pairs of points		
UNDERLINE = 0 1 2 3	Underlines text (0 = no underline, 3 = darkest underline)		

All these options take affect on the text AFTER the option appears. So if we want the first part of the line left justified and the second part right justified, use a command like this:

TITLE JUSTIFY=left 'first part' JUSTIFY=right 'second part';

In our example below, the first change uses the FONT= option to specify a standard font for the entire title (by default, TITLE1 uses Swiss font and the others use the default hardware font). The second change, uses the HEIGHT= option to make the first title the same hieght as the other titles. In order to clearly label the month for this plot, the words 'August 2007' in the second line of the title were highlighted in red. Note that when the title is split in two sections, you must embed the spaces between the sections. SAS will not automatically put a space between them. (See the extra space after the word 'of' in TITLE2 below).

A modification of the footnote was added to separate the two parts of the text. The name of the company was left justified and the date was right justified using the JUSTIFY= option.

```
TITLE FONT='Times New Roman' HEIGHT=1.5

'Number of Hits on Websites 1, 2, and 3';

TITLE2 FONT='Times New Roman' HEIGHT=1.5

'For the Month of ' COLOR=red 'August 2007';

TITLE3 FONT='Times New Roman' HEIGHT=1.5 'Figure 10';

FOOTNOTE JUSTIFY=left 'Company Name'

JUSTIFY=right 'September 15, 2007';

PROC GPLOT DATA=perm.hits;


PLOT Web1 * day = '*'

Web2 * day = '+'

Web3 * day = '-'

/ OVERLAY HREF = '17' FRAME;

RUN;
```



STEP 3 – Adjusting Symbols

PROC GPLOT allows symbols to be specified a couple of ways. The first is similar to the method used in PROC PLOT. However, the '*' symbol produces a different symbol in PROC GPLOT as compared to PROC PLOT. In order to use an asterisk, you must specify the "='star" on the plot statement. Note that there is no way to use a dash as the symbol in PROC GPLOT. If you specify ='dash', you will get a circle with a dot in the middle (included here as an example). See Appendix A for a table of the different symbols available.

```
PROC GPLOT DATA=perm.hits;


PLOT Web1 * day = 'star'

Web2 * day = 'plus'

Web3 * day = '-'

/ OVERLAY HREF = '17' FRAME;

RUN;
```



The SYMBOL Statement

Company Name

Another way to specify the symbol is through the use of the SYMBOL statement. The symbol statement is a very powerful tool providing many other functions besides selecting a symbol. Using the symbol statement, the symbols on the plot can be connected, the types of lines used on the plot can be chosen and the colors of the symbols and lines can be selected. Here are some of the options you can use:


September 15, 2007

OPTION	WHAT IT DOES				
COLOR =	To specify color (also can use CI, CV and CO for colors of the line, points, and outlines, respectively.				
VALUE =	Selects the symbol printed on the plot (see Appendix B for the special symbols available)				
HEIGHT =	To specify size of each symbol on the plot				
INTERPOL	Specifies if you want a line drawn and what type of line to use (also used to ask for regression plots)				
= BOX	For box plots				
= HILO	To draw a single line between min and max values of one axis				
= JOIN	Connect the dots type of line				
= NONE	No line (default)				
others	See SAS documentation for others				
LINE =	Specifies the line to use (see Appendix B)				
WIDTH =	Specifies the thickness of the line				
POINTLABEL =	Labels the plot points				

Symbol statements are numbered 1 through 99 and are used consecutively for each combination of plot variables. In this example, we have three separate plots (web1*day, web2*day and web3*day). If the symbol statement does not specify a color, then the SAS system uses that symbol statement repeatedly through all the colors in the color palette before going to the next symbol statement. It is therefore recommended that color be specified in each symbol statement so that the user is not just relying only on the symbol color to determine the meaning of the plot line or symbol. Color can be specified for both the line and the symbol (COLOR=), or can be specified separately for each (CL=line color and CV=symbol color). For this example, three symbol statements, each with a different color, were used.

To select the symbols, the VALUE= (V=) option is used. Refer to the SAS documentation for a table of the symbols available. To choose the size of the symbol, use the HEIGHT= (H=) option. Default height is 1.


SYMBOL1 COLOR=blue VALUE=dot H=1; SYMBOL2 COLOR=red VALUE=star H=1; Symbol3 COLOR=green VALUE=circle H=1; PROC GPLOT DATA=perm.hits; PLOT Web1*day Web2*day Web3*day / OVERLAY HREF = '17' FRAME; RUN;



To specify that the points be connected, use the INTERPOL= (or I=) option. This feature offers many variations on connecting the symbols. For example, to use a straight "connect the dots" type of line, use 'I=join', or to get a regression line use 'I=R', or to get a step line use 'I=needle'. If you do not wish to connect the symbols, choose I=none. For this plot, a straight line between the points makes the plot much easier to understand, so 'I=join' was used.

To select what type of line to use, use the LINE= (L=) option. See Appendix B for a list of the lines that may be used. There are 46 different lines that can be selected. Line Type 1 is a solid line and Line Type 0 is no line. All the others are combinations of dashed lines of different sized dashes. To choose the thickness of the line use the Width= (W=) option. The default is 1.


SYMBOL1 COLOR=blue INTERPOL=join LINE=1 VALUE=dot; SYMBOL2 COLOR=red INTERPOL=join LINE=2 VALUE=star; SYMBOL3 COLOR=green INTERPOL=join L=3 VALUE=circle; PROC GPLOT DATA=perm.hits; PLOT Web1*day Web2*day Web3*day / OVERLAY HREF = '17' FRAME; RUN;



STEP 4 - Adjusting the Legend

Using overlay with the PROC GPLOT procedure does not automatically produce a legend. To produce a legend, use the LEGEND option in the Plot statement.

PROC GPLOT DATA=perm.hits;
PLOT Web1*day Web2*day Web3*day
/ OVERLAY HREF = '17' FRAME **LEGEND**;
RUN;



The LEGEND Statement

Note that in the plot above, the label for Web1 displays in the legend rather than the variable name. It is possible to specify your own legend labels using the LEGENDx statement and the VALUE= option. In the LEGENDx statement, the 'x' is a number from 1 to 99, which labels the different legend statements. Within the VALUE= option you can specify your own text, justify it, and/or specify different portions of the text different colors, fonts, or height. To change the name of the legend itself (in the example above, the default label was 'PLOT'), we can use the LABEL= option.

LEGEND statements specify the characteristics of the legend, but do not create the legend. The characteristics that can be specified include the position and appearance of the legend box, the text and appearance of the legend label, the appearance of the legend entries, including the size and shape of the legend values, and the text of the labels for the legend values.

OPTION	WHAT IT DOES			
LABEL = (text edit options)	Modify/specify the legend title			
VALUE = (text edit options)	Modify the legend value descriptions			
SHAPE = BAR LINE SYMBOL	Specifies the size and shape of the legend values			
POSITION =	Positions the legend on the graph.			
(<bottom middle="" top="" =""></bottom>	By default, POSITION=(Bottom Center Outside)			
<left center="" right="" =""></left>				
<outside inside="" ="">)</outside>				
MODE = Protect Reserve	Defines how the legend interacts with the graphing area.			
Share				
ORDER =	Orders the legend values			
ACROSS =	Specifies how legend entries will appear			
DOWN =				
FRAME =	Draws a frame around the legend			
FWIDTH =	Specifies the thickness of the frame			
OFFSET = (x,y)	Moves the legend			
CBLOCK =, CSHADOW =,	Specifies color of various parts of the legend or to get shadow			
CBORDER =, CFRAME	effects.			

Text Description Suboptions can be used whenever you want to modify the appearance of text (inside the label and value options, for example). Text description suboptions affect all the strings that follow unless the suboption is changed or turned off.

OPTION	WHAT IT DOES		
COLOR =	Specifies color		
FONT =	Specifies font		
HEIGHT =	Specifies size of font		
JUSTIFY = LEFT CENTER RIGHT	Justifies portions of the text		
POSITION =	Places the legend label in relation to the legend entries.		
(<bottom middle="" ="" top=""></bottom>	Used only with the LABEL option.		
<left center="" right="" ="">)</left>	By default, POSITION=LEFT.		
TICK =	Modifies one legend engtry		

Unlike the symbol statements, which are automatically used as the plotting process needs them, you will need to explicitly specify which legend statement to use for your plot. To do this, specify legend=legendx (where x is which legend statement you are referring to) on the plot statement.

LEGEND1 LABEL=(HEIGHT=1 'Web Sites')


VALUE=('Web Site 1' 'Web Site 2' 'Web Site 3');

PROC GPLOT DATA=perm.hits;

PLOT Web1*day Web2*day Web3*day

/ OVERLAY HREF = '17' FRAME LEGEND=legend1;

RUN:



In addition to specifying the text for the legend entries, the entries may be repositioned within the legend itself. Using the ACROSS= and DOWN= options, I specified that the legend should be listed vertically rather than horizontally.

The legend may also be moved either inside or outside (default) the plotting area using the POSITION= option. If the legend is placed outside the plot, this will limit how much plotting space is actually given to the plot. If you wish to save plotting space, move the legend inside the plotting area. The default position is (Bottom Center Outside). I chose to move it to (Top Left Inside).

If the legend is moved to the inside of the plotting area, the option MODE must be changed to either 'PROTECT' or 'SHARE'. With Protect, the legend covers any points that would otherwise appear in the same place as the legend. With Share, the legend and the points are both printed (on top of each other). For this example, PROTECT was used.

To move the legend label to the top of the list of legend entries, use the LABEL= option and its suboptions POSITION and JUSTIFY to move it above the entries and center it.

```
LEGEND1 LABEL=(HEIGHT=1 POSITION=top JUSTIFY=center 'Web Sites')

VALUE=('Web Site 1' 'Web Site 2' 'Web Site 3')

ACROSS=1 DOWN=3

POSITION= (top left inside)


MODE=protect;

PROC GPLOT DATA=perm.hits;

PLOT Web1*day Web2*day Web3*day

/ OVERLAY HREF = '17' FRAME LEGEND=legend1;

RUN;
```



Two additional options to make the legend stand out are FRAME, to add a frame around the legend and CFRAME to shade it. These two options are mutually exclusive in GPLOT. To add a frame, use the FRAME option in the legend statement. To shade the legend, use the CFRAME option and specify the color you want. I preferred the CFRAME option for this example.

In order to improve the placement of the legend, it was moved slightly to the right rather than against the vertical axis. To move the legend to the right, use the option OFFSET=(x,y). The x specifies the amount of space to move horizontally, and y specifies the amount of space to move the legend vertically. With the OFFSET option, you can specify the units you want used for the x and y values. I chose to use percent of the plotting space (pct). You can also specify to use inches (in), or centimeters (cm). If the x or y value is not specified, it is assumed to be zero.

```
LEGEND1 LABEL=(HEIGHT=1 POSITION=top JUSTIFY=center
'Web Sites')

VALUE=('Web Site 1' 'Web Site 2' 'Web Site 3')

ACROSS=1 DOWN=3

POSITION = (top left inside)

MODE=protect

OFFSET = (3 pct)


CFRAME = yellow;


PROC GPLOT DATA=perm.hits;

PLOT Web1*day Web2*day Web3*day

/ OVERLAY HREF = '17' FRAME LEGEND=legend1;

RUN;
```







STEP 5 – The Axes

The axes can be modified in two ways. You may use the options on the PLOT statement in the GPLOT procedure or you may use the AXIS statement and then assign the particular AXIS definition in the PLOT statement.

In this example, modifying the vertical axis so there are fewer numbers displayed will make the plot look cleaner and easier to read. To do this use the VAXIS= option. Since these numbers were numeric it is possible to specify the axis using a range. During a test run, I noticed SAS placed four tick marks between each major tick mark. I thought that three would look better and be easier to understand (to represent every 50 hits). To do this, use the VMINOR= option.

PROC GPLOT DATA=perm.hits;
PLOT Web1*day Web2*day Web3*day
/ OVERLAY HREF = '17' FRAME LEGEND=legend1
VAXIS=0 to 1800 by 200 VMINOR=3;
Run;





The AXIS Statement

The second way to modify the axes is to use the AXIS statement. Here we are using it to modify the placement of the axis labels (not available in the options on the plot statement). The AXIS statement is similar to the LEGEND and SYMBOL statements. You can specify up to 99 different AXIS statements by specifying AXISx where x can be any number from 1 to 99. To choose which axis definition to use, use the VAXIS= and/or the HAXIS= options on the PLOT statement. This will replace the range of numbers in the previous example.


AXIS statements specify the characteristics of an axis, including the way the axis is scaled, how the data values are ordered, the location and appearance of the axis lines and tick marks, and the text and appearance of the axis label and major tick marks. Many of the options that can be used in the AXIS statement are the same as the TITLE statement options (i.e., ANGLE and ROTATE suboptions).

OPTION	WHAT IT DOES		
COLOR =	Color of the axes		
LENGTH =	Specify axis length		
STYLE =	Specify line type to use for the axis		
WIDTH =	Specify the thickness of the axis line		
MAJOR = (tick mark suboptions)	Specify the major tick marks		
MINOR = (tick mark suboptions)	Specify the minor tick marks		
ORDER =	Value list		
LABEL = (text arguments)	Specify the label on the axis		
REFLABEL = (text arguments)	Add a label to a reference line on the plot		
VALUE = (text arguments)	Specify replacement values for the tick marks		

Use the ANGLE and ROTATE suboptions in the LABEL option to change the angle of the vertical axis label to display it down the side instead of across the top. Note the suboptions on the AXIS statement are applied to any text that comes *after* the options. If you put the text before the suboptions, the suboptions will be ineffective. The ANGLE suboption causes the entire label to be turned. The ROTATE suboption will turn each letter separately. Note: the direction of –90 degrees will turn the same amount as specifying 270 degrees. The size of the label was increased using the HEIGHT suboption in the label option.

In this example, two changes were made to the horizontal axis. First, a label was added to the reference line, using the REFLABEL= option. Second, removal of the leading zeroes from the tick mark labels for the 1st through the 9th of August using the VALUE= option. Note that these values are character, so the text must be specified in quotes and a range could not be used (as is possible with numeric variables). The SAS system assigns the values from the first point consecutively for as many labels as provided, and then uses the default label for the rest of the points.

Number of Hits on Websites 1, 2, and 3 For the Month of August 2007 Figure 19



Company Name September 15, 2007

FINAL PROC GPLOT PROGRAM

```
LEGEND1 LABEL=(HEIGHT=1 POSITION=top JUSTIFY=center
           'Web Sites')
      VALUE=('Web Site 1' 'Web Site 2' 'Web Site 3')
      ACROSS=1 DOWN=3 MODE=protect
      POSITION = (top left inside)
      OFFSET = (3 pct) CFRAME = yellow;
AXIS1
          LABEL=(angle=270 ROTATE=90 HEIGHT=1.5
              'Number of Hits')
     ORDER=(0 to 1800 by 200)
     MINOR=(number=3);
AXIS2 REFLABEL=(POSITION=top JUSTIFY=center 'Email Ad')
     VALUE=('1' '2' '3' '4' '5' '6' '7' '8' '9');
SYMBOL1 COLOR=blue INTERPOL=join LINE=1 VALUE=dot;
SYMBOL2 COLOR=red INTERPOL=join LINE=2 VALUE=star;
SYMBOL3 COLOR=green INTERPOL=join LINE=3 VALUE=circle;
TITLE FONT='Times New Roman' HEIGHT=1.5
       'Number of Hits on Websites 1, 2, and 3':
TITLE2 FONT='Times New Roman' HEIGHT=1.5
       'For the Month of ' COLOR=red 'August 2007';
FOOTNOTE JUSTIFY=left 'Company Name'
   JUSTIFY=right 'September 15, 2007';
PROC GPLOT DATA=perm.hits;
   PLOT Web1*day Web2*day Web3*day
      / OVERLAY HREF = '17' FRAME LEGEND=legend1
      VAXIS=axis1 HAXIS=axis2;
RUN:
```

CONCLUSION AND TRADEMARKS

The options described here are by no means exhaustive of what the SAS GRAPH system can do. Please refer to the documentation to learn what else is possible to improve the quality of your graphs.

AUTHOR CONTACT

Your comments and questions are valued and welcome. Contact the author at:

Wendi L. Wright 1351 Fishing Creek Valley Rd. Harrisburg, PA 17112 Phone: (717) 513-0027

E-mail: wendi_wright@ctb.com

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.

Appendix A
Special Symbols for Plotting Data Points

VALUE=	Plot Symbol	VALUE=		Plot Symbol
PLUS	+	%	(percent)	දු
×	\times	&	(ampersand)	\$
STAR	*	١.	(single quote)	₩
SQUARE		=	(equals)	$\stackrel{\wedge}{\sim}$
DIAMOND	\Diamond	-	(hyphen)	\odot
TRIANGLE	\triangle	@	(at)	₽
HASH	\sharp		(asterisk)	2
Y .	Y	+	(plus)	\oplus
z	Z	>	(greater than)	o'
PAW			(period)	24
POINT		<	(less than)	h
DOT	•	,	(comma)	ô
CIRCLE	0	/	(slash)	Ψ
_ (underscore)		?	(question mark)	P
" (double quote)	\Diamond	((left parenthesis)	\mathbb{C}
# (pound sign)	\bigcirc)	(right parenthesis)	G
\$ (dollar sign)	\Diamond	:	(colon)	*

Appendix B

Line Types

